

# Verilog-AMS

## 実用モデリング講座

第4回/最終回

### テストベンチとモデリングの落とし穴



桜井 至, 石塚総一郎


前回(本誌2003年6月号, pp.138-144)は, イベント・ベース・モデルを記述する方法, およびPLL回路やサンプル・ホールド回路の記述例を紹介しました. 今回はVerilog-AMSのテストベンチなどについて解説します. (編集部)

Verilog-AMSを用いて, テストベンチで使用する機能モジュールを記述することができます(図1). 今回はこれらのモジュールについて解説します. また, 最終回のまとめとして, モデリングの際に陥りやすい落とし穴やVerilog-AMSの今後についても述べます.

#### ●被測定回路に与える信号を生成するスティミュラス

SPICEシミュレータでは, 通常, AC信号源やDC信号源などを組み合わせて信号を生成し, 測定対象となる回路(被測定回路)に入力します. この信号生成を行う回路を, 一般にスティミュラス・モジュールと呼びます.

Verilog-AMSを用いると, スティミュラス・モジュールについて, より詳細な動作の記述や自由度の高い記述を行うことができます<sup>注1</sup>. 以下に, FM信号源, 離散正弦波信号源, PRBS(疑似乱数ビット列)源の記述を紹介します.



〔図1〕スティミュラス・モジュールと測定モジュール  
OPアンプの過渡特性の測定回路を示している。

#### 1) FM信号源

初めに, Verilog-AMSを用いてSPICE3のSFFM(単一周波数FM信号源)を記述します. SFFMは, 以下の式で表現されます<sup>注2</sup>.

$$V(t) = V_o + V_a \sin(2\pi F_c t + MDI \sin(2\pi F_s t)) \quad (1)$$

ここで,  $V_o$ はオフセット電圧,  $V_a$ は振幅,  $F_c$ は搬送周波数,  $MDI$ は変調指数,  $F_s$ は信号周波数です. SPICE3でインスタンス化するには, 以下のように記述します.

```
V1 N1 N2 SFFM ( Vo Va Fc MDI Fs )
```

Verilog-AMSによる記述をリスト1に示します. 式(1)がほぼそのまま, analog begin ~ endの間に記述されています. また, バウンド・ステップ関数により, 時間ステップを制御しています.

#### 2) 離散正弦波信号源

A-Dコンバータの構成部品の性能を確認する場合, 離散的な入力信号を必要とします. リスト2に示される離散正弦波信号源は, 通常正弦波をタイム関数によってサンプルして, 離散的な正弦波を生成しています. 動作のようすを図2に示します.

#### 3) PRBS(疑似乱数ビット列)源

PRBSは信号伝送品質の評価基準であるビット・エラー率(BER: bit error rate)の測定などに利用されます. リスト3に記述例を示します. \$rdist\_uniform関数<sup>注3</sup>によ

注1: Verilog-AMSシミュレータに組み込まれていない信号源も自由に作成できる.

注2: SPICEシミュレータではこの式がFORTRANやC言語で記述され, 組み込まれている.

注3: 過去のVerilog-AMS LRM(Language Reference Manual)では\$dist\_uniformとして規定されている.

〔リスト1〕 FM 信号源の記述例

```

`include "disciplines.h"
`include "constants.h"
module sffm_vams(inp, inn);
  inout inp, inn;
  electrical inp, inn;

  parameter real Vo = 0.0;      // オフセット
  parameter real Va = 0.2;      // 振幅
  parameter real Fc = 100M;     // 搬送周波数
  parameter real MDI = 0;       // 変調指数
  parameter real Fs = 1M;       // 信号周波数

  analog begin
    $bound_step(1/Fs/10);      // バウンド・ステップ関数
    V(inp, inn) <+ Vo
      + Va * sin(`M_TWO_PI * Fc * $abstime
        +MDI*sin(`M_TWO_PI * Fs * $abstime ))
  end
endmodule

```

〔リスト2〕 離散正弦波信号源の記述例

```

`include "disciplines.h"
`include "constants.h"
module samplesine(outp, outn);
  inout outp, outn;
  electrical outp, outn;

  parameter real Vo = 0.0;      // オフセット
  parameter real Va = 0.2;      // 振幅
  parameter real Fc = 100e6;    // 信号周波数
  parameter real Fs = 90e6;     // サンプリング周波数
  parameter real Td = 0;        // 出力遅延時間
  real trf, ssin;

  analog begin
    @(initial_step)
      trf = 1/Fs/100;           // 立ち上がり時間, 立ち下がり時間
    @(timer(1/Fs, 1/Fs))
      ssin = sin(`M_TWO_PI * Fc * $abstime);
    V(outp, outn) <+ Vo + Va * transition(ssin, Td, trf);
  end
endmodule

```

り乱数 newedge を生成し、1 周期 (period 時間) ごとに newedge を 1 ずつ減じます。newedge が 0 になると出力を反転し、新たに乱数を生成します。パラメータ min\_run, max\_run はそれぞれ出力が変化しない最小、最大の周期数を決定します。これにより、min\_run, max\_run の範囲内でランダムにエッジが変化する信号を生成できます (図3)。

### ●出力信号を取り出して見やすく加工する測定モジュール

測定モジュールは被測定回路から出力される信号を観測し、測定に必要な処理を施す回路です。Verilog-AMS を利

〔リスト3〕 PRBS の記述例


```

`include "disciplines.h"
module prbs_diff_source(outp, outn, common);
  inout outp, outn, common;
  electrical outp, outn, common;

  parameter real vcm = 0;       // 出力コモン・モード電圧
  parameter real vswing = 1.0; // 出力振幅
  parameter integer seed = 123; // 乱数のシード
  parameter real period = 1n;  // 周期
  parameter integer min_run = 1; // 同一出力の最小連続回数
  parameter integer max_run = 10; // 同一出力の最大連続回数
  parameter real trf = -1; // 立ち上がり時間, 立ち下がり時間
  real rft;
  integer seed2, newedge, outstate;


  analog begin
    @(initial_step) begin
      if(trf < 0 || trf > period/2)
        //trf<0 であれば立ち上がり, 立ち下がりは period/10
        rft = period/10;
      else
        rft = trf; //そうでなければ指定された値
      seed2 = seed;
      newedge = 0;
    end
    @(timer(period, period)) begin
      if(newedge == 0) begin
        newedge = $rdist_uniform(seed2, min_run,
          max_run); //乱数生成
        outstate = !outstate;
      end
      newedge = newedge - 1;
    end
    end
    V(outp) <+ vcm - vswing/2 + vswing *
      transition(outstate, 0.0, rft);
    V(outn) <+ vcm - vswing/2 + vswing *
      transition(!outstate, 0.0, rft);
  end
endmodule

```



〔図2〕 離散正弦波の信号源の動作波形

リスト2の記述について、シミュレーションした結果の波形を示している。



〔図3〕 PRBS の動作波形

リスト3の記述について、シミュレーションした結果の波形を示している。